clear all
set matsize 5000

set more off
use "${datapath}/fascism_db.dta",clear

* standardize dependent and independent vars
foreach var of varlist fascist_violence-strike_agr_1920 lpop1911-volunteers1866_pop rain1918-rain_var {
egen z_`var'=std(`var')

}

* relabel them
run "labels.do"


*****
local demo "z_lpop1911 z_lpop2 z_lpop3 z_lpop4 z_sh_pop_1911_be6"
local cntls_spec1 "z_lpop1911 z_lpop2 z_lpop3 z_lpop4 z_sh_pop_1911_be6 d_reg* province*"
local cntls_spec2 "`cntls_spec1' z_larea z_centre_alt z_max_alt"
local cntls_spec3 "`cntls_spec2' z_psu1913_vv"
local cntls_spec4 "`cntls_spec3' z_veterans74_95 z_veterans96_00 z_ard_vol_pop6m z_cruent_d z_army_suppliers_d"
local cntls_spec5 "`cntls_spec4' z_dlab z_shcrop z_landlord_ass"
local cntls_spec6 "`cntls_spec5' z_ind_workers z_ind_firms z_literacy z_elites z_bourgeoisie"
***Specifications with Alternative Fixef Effects
local sem_fe "d_fr1_d_ds_may1915-d_fr16_d_ds_may1918"
local cntls_alt_sem_fe_spec1 "`demo' `sem_fe' province*"
local cntls_alt_sem_fe_spec4 "`cntls_alt_sem_fe_spec1' z_larea z_centre_alt z_max_alt z_psu1913_vv z_veterans74_95 z_veterans96_00 z_ard_vol_pop6m z_cruent_d z_army_suppliers_d"
local cntls_alt_sem_fe_spec6 "`cntls_alt_sem_fe_spec4' z_dlab z_shcrop z_landlord_ass z_ind_workers z_ind_firms z_literacy z_elites z_bourgeoisie"
***
local month_fe "d_fr1_d_dm_may1915-d_fr16_d_dm_nov1918"
local cntls_alt_month_fe_spec1 "`demo' `month_fe' province*"
local cntls_alt_month_fe_spec4 "`cntls_alt_month_fe_spec1' z_larea z_centre_alt z_max_alt z_psu1913_vv z_veterans74_95 z_veterans96_00 z_ard_vol_pop6m z_cruent_d z_army_suppliers_d"
local cntls_alt_month_fe_spec6 "`cntls_alt_month_fe_spec4' z_dlab z_shcrop z_landlord_ass z_ind_workers z_ind_firms z_literacy z_elites z_bourgeoisie"
***Specifications For Excess Mortality (without regiment fixed effects)
local cntls_excess_spec1 "`demo' province*"
local cntls_excess_spec4 "`cntls_excess_spec1' z_larea z_centre_alt z_max_alt z_psu1913_vv z_veterans74_95 z_veterans96_00 z_ard_vol_pop6m z_cruent_d z_army_suppliers_d"
local cntls_excess_spec6 "`cntls_excess_spec4' z_dlab z_shcrop z_landlord_ass z_ind_workers z_ind_firms z_literacy z_elites z_bourgeoisie"


*APPENDIX

*TAB_A3 - NO FE
reg z_psu1919_vv z_foots_pop6m `demo', cl(`clustering'circondario1921)
outreg2 using "${outpath}/TAB_A3.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster replace
*
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv {
qui ivreg2 `depvar' (z_psu1919_vv=z_foots_pop6m) `demo', cl(circondario1921)
outreg2 using "${outpath}/TAB_A3.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters",e(N_clust),"First-stage F-stat",e(widstat)) nocons label noaster append
}

*TAB_A4 ALTERNATIVE FIRST STAGES
foreach instr of varlist z_draftres_foots_pop6m z_draft_foots_pop6m z_casualties_pop6m{
foreach i of numlist 1 4 6 {
qui reg z_psu1919_vv `instr' `cntls_spec`i'', cl(circondario1921)
qui test `instr'
outreg2 using "${outpath}/TAB_A4.tex", dec(2) keep(`instr') addstat("Number of clusters", e(N_clust), 1st stage F-stat,r(F)) nocons label noaster append
}
}
*

*TAB_A5 (REDUCED FORM FOR THE FOUR MAIN OUTPUT VARIABLES)
foreach depvar in z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv {
foreach i of numlist 1 4 6 {
qui reg `depvar' z_foots_pop6m `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A5.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
}


*TAB A6 ON TOTAL EFFECTS OF MILITARY CONTROLS
*PANEL A
foreach i of numlist 4 6 {
qui reg z_psu1919_vv z_foots_pop6m `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A6A.tex", dec(2) keep(z_veterans74_95 z_veterans96_00 z_ard_vol_pop6m z_army_suppliers_d) nocons label noaster append
}
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv{
foreach i of numlist 4 6 {
qui ivreg2 `depvar' (z_psu1919_vv=z_foots_pop6m) `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A6A.tex", dec(2) keep(z_veterans74_95 z_veterans96_00 z_ard_vol_pop6m z_army_suppliers_d) nocons label noaster append
}
}

*PANEL B 
matrix Table_A6B = J(8,10,.)

**store first stage
foreach i of numlist 4 6 {
qui reg z_psu1919_vv z_foots_pop6m `cntls_spec`i''
predict z_psu_hat`i',xb
est store first`i'
}
**store first stage for Fascist vote share in 1921
foreach i of numlist 4 6 {
qui reg z_psu1919_vv z_foots_pop6m `cntls_spec`i'' if z_fascist1921_vv!=.
predict z_psu_hat`i'b,xb
est store first`i'b
}
**store IV
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv{
foreach i of numlist 4 6 {
if `depvar'==z_fascist1921_vv {
qui reg `depvar' z_psu_hat`i'b `cntls_spec`i''
est store `depvar'`i'
}
else {
qui reg `depvar' z_psu_hat`i' `cntls_spec`i''
est store `depvar'`i'
}
}
}
local x=2
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv{
local x=`x'+2
foreach i of numlist 4 6{
if `depvar'==z_fascist1921_vv suest `depvar'`i' first`i'b, cl(circondario1921)
else suest `depvar'`i' first`i', cl(circondario1921)

foreach military of varlist z_veterans74_95 z_veterans96_00 z_ard_vol_pop6m z_army_suppliers_d {
local y=`y'+2
if `depvar'==z_fascist1921_vv nlcom [`depvar'`i'_mean]_b[`military']+[`depvar'`i'_mean]_b[z_psu_hat`i'b]*[first`i'b_mean]_b[`military']
else nlcom [`depvar'`i'_mean]_b[`military']+[`depvar'`i'_mean]_b[z_psu_hat`i']*[first`i'_mean]_b[`military']
matrix b = r(b)
matrix V = r(V)
local se sqrt(V[1,1])
local se_str "(`se')"
matrix Table_A6B[`y'-1,`x'-1+`i'/2-2] = b[1,1]
matrix Table_A6B[`y',`x'-1+`i'/2-2]= `se'
}
local y=0
}
}
matrix rownames Table_A6B= "Veterans (classes 1874-1895)" " " "Veterans (classes 1896-1900)" " " "Assault tr. + volunt. casualties" " "  "Army supplying production plant" " "
matrix colnames Table_A6B= "Fascist violence in 1920-2" "Fascist violence in 1920-2" "Fascist branches in 1921" "Fascist branches in 1921" "Fascist vote share in 1921" "Fascist vote share in 1921" "Fascist vote share in 1924" "Fascist vote share in 1924"

estout matrix(Table_A6B ,fmt(2)) using "${outpath}/TAB_A6B.tex", style(tex) replace



*TAB_A7 (additional first stage for fascist1921_vv
foreach i of numlist 1/6 {
qui reg z_psu1919_vv z_foots_pop6m `cntls_spec`i'' if z_fascist1921_vv!=., cl(circondario1921)
qui test z_foots_pop6m
outreg2 using "${outpath}/TAB_A7.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust),"1st stage F-stat",r(F)) nocons label noaster append
}

*TAB_A8 - No South
foreach i of numlist 1 4 6 {
qui reg z_psu1919_vv z_foots_pop6m `cntls_spec`i'' if no_south==1, cl(circondario1921)
outreg2 using "${outpath}/TAB_A8.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
foreach depvar in z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv {
foreach i of numlist 1 4 6 {
qui ivreg2 `depvar' (z_psu1919_vv=z_foots_pop6m) `cntls_spec`i'' if no_south==1, partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A8.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust) , "1st stage F-stat", e(widstat)) nocons label noaster append
}
}
*TAB_A9 second stage alternative IV
*PANEL A
foreach instr of varlist z_draftres_foots_pop6m z_draft_foots_pop6m z_casualties_pop6m{
foreach i of numlist 1 4 6 {
qui ivreg2 z_fascist_violence (z_psu1919_vv=`instr') `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A9A.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust)) nocons label noaster append
*PANEL B
qui ivreg2 z_fascist_branch (z_psu1919_vv=`instr') `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A9B.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust)) nocons label noaster append
*PANEL C
qui ivreg2 z_fascist1921_vv (z_psu1919_vv=`instr') `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A9C.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust)) nocons label noaster append
*PANEL D
qui ivreg2 z_fascist1924_vv (z_psu1919_vv=`instr') `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A9D.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
}

	
*TAB_A10 alt_fe
foreach j in sem_fe month_fe {
foreach i of numlist 1 4 6 {
qui ivreg2 z_fascist_violence (z_psu1919_vv=z_foots_pop6m) `cntls_alt_`j'_spec`i'', partial(``j'' province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A10A.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
qui ivreg2 z_fascist_branch (z_psu1919_vv=z_foots_pop6m) `cntls_alt_`j'_spec`i'', partial(``j'' province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A10B.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
qui ivreg2 z_fascist1921_vv (z_psu1919_vv=z_foots_pop6m) `cntls_alt_`j'_spec`i'', partial(``j'' province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A10C.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
qui ivreg2 z_fascist1924_vv (z_psu1919_vv=z_foots_pop6m) `cntls_alt_`j'_spec`i'', partial(``j'' province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A10D.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
}


*TAB_A11
*Panel A
foreach depvar of varlist z_fascist_murders z_political_violence z_other_violence {
foreach i of numlist 1 4 6 {
qui ivreg2 `depvar' (z_psu1919_vv=z_foots_pop6m) `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A11A.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust) , "1st stage F-stat", e(widstat)) nocons label noaster append
}
}
*Panel B
foreach depvar of varlist z_fascist1921restr z_pnf1921_vv z_pnf1921_d{
foreach i of numlist 1 4 6 {
qui ivreg2 `depvar' (z_psu1919_vv=z_foots_pop6m) `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A11B.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust) , "1st stage F-stat", e(widstat)) nocons label noaster append
}
}
*Panel C
foreach depvar of varlist z_fascist1921_vv z_fascist1924_vv{
foreach i of numlist 1 4 6 {
qui ivreg2 `depvar' (z_psu1919_vv=z_foots_pop6m) z_fascist1919_vv `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A11C.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust) , "1st stage F-stat", e(widstat)) nocons label noaster append
}
}
*
foreach i of numlist 1 4 6 {
qui ivreg2 z_off_fascist1924_vv (z_psu1919_vv=z_foots_pop6m) `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A11C.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust) , "1st stage F-stat", e(widstat)) nocons label noaster append
}



*TAB_A12 RED SCARE SOCIALIST 1920 & INDEX
*panel A
foreach i of numlist 1 4 6 {
qui reg z_psu1920d z_foots_pop6m `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A12A.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv{
foreach i of numlist 1 4 6 {
qui ivreg2 `depvar' (z_psu1920d=z_foots_pop6m) `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A12A.tex", dec(2) keep(z_psu1920d) addstat("Number of clusters", e(N_clust), "1st Stage F-stat", e(widstat)) nocons label noaster append
}
}
*panel B
foreach i of numlist 1 4 6 {
qui reg z_red_scare_index z_foots_pop6m `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A12B.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv{
foreach i of numlist 1 4 6 {
qui ivreg2 `depvar' (z_red_scare_index=z_foots_pop6m) `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A12B.tex", dec(2) keep(z_red_scare_index) addstat("Number of clusters", e(N_clust), "1st Stage F-stat", e(widstat)) nocons label noaster append
}
}
*panel C
foreach i of numlist 1 4 6 {
qui reg z_psu1921_vv z_foots_pop6m `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A12C.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv{
foreach i of numlist 1 4 6 {
qui ivreg2 `depvar' (z_psu1921_vv=z_foots_pop6m) `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A12C.tex", dec(2) keep(z_psu1921_vv) addstat("Number of clusters", e(N_clust), "1st Stage F-stat", e(widstat)) nocons label noaster append
}
}

*TAB A13: NEVER-TAKERS
reg z_psu1919_vv z_foots_pop6m `cntls_spec1'
predict psu1919hat, xb

sum psu1919hat,d
g nt1919 = psu1919hat<r(p25)
g nt1913 = psu_cand1913==0


foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv {
foreach i of numlist 1 4 6 {
qui reg `depvar' z_foots_pop6m `cntls_spec`i'', cluster(circondario1921)


preserve
clear
matrix res = r(table)
svmat res, names("V") 
keep V1 // keep only 1st coef stats
keep in 4 // keep only pvalue
rename V1 standard_pv 
gen depvar = "`depvar'"
gen spec = `i'
gen sample = "Full"
tempfile FSpv_`depvar'_spec`i'
save `FSpv_`depvar'_spec`i''
restore


outreg2 using "${outpath}/TAB_A13A.tex", dec(2) keep(z_foots_pop6m) nocons label noaster append
eststo: qui reg `depvar' z_foots_pop6m `cntls_spec`i''
***predicted socialist in 1919 in the lowest quartile
eststo: qui reg `depvar' z_foots_pop6m `cntls_spec`i'' if nt1919==1


qui suest est1 est2, cluster(circondario1921)
test [est1_mean]z_foots_pop6m=[est2_mean]z_foots_pop6m
local pvalue=r(p)
qui reg `depvar' z_foots_pop6m `cntls_spec`i'' if nt1919==1, cluster(circondario1921)

preserve
clear
matrix res = r(table)
svmat res, names("V") 
keep V1 // keep only 1st coef stats
keep in 4 // keep only pvalue
rename V1 standard_pv
gen depvar = "`depvar'"
gen spec = `i'
gen sample = "NT19"
tempfile NT19pv_`depvar'_spec`i'
save `NT19pv_`depvar'_spec`i''
restore

outreg2 using "${outpath}/TAB_A13B.tex", dec(2) keep(z_foots_pop6m) addstat(P-value, `pvalue') nocons label noaster append
***no socialist candidate in 1913
eststo: qui reg `depvar' z_foots_pop6m `cntls_spec`i'' if nt1913==1

qui suest est1 est3, cluster(circondario1921)
test [est1_mean]z_foots_pop6m=[est3_mean]z_foots_pop6m
local pvalue=r(p)
qui reg `depvar' z_foots_pop6m `cntls_spec`i'' if nt1913==1, cluster(circondario1921)

preserve
clear
matrix res = r(table)
svmat res, names("V") 
keep V1 // keep only 1st coef stats
keep in 4 // keep only pvalue
rename V1 standard_pv
gen depvar = "`depvar'"
gen spec = `i'
gen sample = "NT13"
tempfile NT13pv_`depvar'_spec`i'
save `NT13pv_`depvar'_spec`i''
restore

outreg2 using "${outpath}/TAB_A13C.tex", dec(2) keep(z_foots_pop6m) addstat(P-value, `pvalue') nocons label noaster append
eststo clear
}
}

preserve
use `FSpv_z_fascist_violence_spec1', clear
append using `FSpv_z_fascist_violence_spec4'
append using `FSpv_z_fascist_violence_spec6'

foreach depvar in z_fascist_branch z_fascist1921_vv z_fascist1924_vv {
foreach i in 1 4 6 {

append using `FSpv_`depvar'_spec`i''

}
}

foreach depvar in z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv {
foreach i in 1 4 6 {

append using `NT13pv_`depvar'_spec`i''
append using `NT19pv_`depvar'_spec`i''

}
}
* export pvalues to adjust them to FWER
export delimited using "${outpath}/unadj_pv_sp146_FLS_NT13NT19.csv", replace
restore


*TAB_A15: other parties in 1921 and 1924 controlling for 1919
*Panel A
foreach var in ppi traditional {
foreach i of numlist 1 4 6 {
qui ivreg2 z_`var'1921_vv (z_psu1919_vv=z_foots_pop6m) z_`var'1919_vv `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A15A.tex", dec(2) keep(z_psu1919_vv z_`var'1919_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
}
*Panel B
foreach var in ppi traditional {
foreach i of numlist 1 4 6 {
qui reg z_`var'1921_vv z_foots_pop6m z_`var'1919_vv `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A15B.tex", dec(2) keep(z_foots_pop6m z_`var'1919_vv) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
}
*Panel C
foreach var in ppi traditional {
foreach i of numlist 1 4 6 {
qui ivreg2 z_`var'1924_vv (z_psu1919_vv=z_foots_pop6m) z_`var'1919_vv `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A15C.tex", dec(2) keep(z_psu1919_vv z_`var'1919_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
}
*Panel D
foreach var in ppi traditional {
foreach i of numlist 1 4 6 {
qui reg z_`var'1924_vv z_foots_pop6m z_`var'1919_vv `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A15D.tex", dec(2) keep(z_foots_pop6m z_`var'1919_vv) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
}

*TAB_A16 Large Donors
g interaction=psu1919_vv*landlord_ass
egen z_interaction=std(interaction)
la var z_interaction "Socialists x elite variable"
g IV_interaction=foots_pop6m*landlord_ass
egen z_IV_interaction=std(IV_interaction)
foreach i of numlist 1 4 6 {
ivreg2 z_large_donors (z_psu1919_vv z_interaction =z_foots_pop6m z_IV_interaction) z_landlord_ass `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A16.tex", dec(2) keep(z_psu1919_vv z_interaction) addstat("Number of clusters", e(N_clust), "Kleibergen-Paap F-stat", e(widstat)) nocons label noaster append
}
drop interaction IV_interaction z_interaction z_IV_interaction
***
g interaction=psu1919_vv*z_elites
egen z_interaction=std(interaction)
la var z_interaction "Socialists x elite variable"
g IV_interaction=foots_pop6m*z_elites
egen z_IV_interaction=std(IV_interaction)
foreach i of numlist 1 4 6 {
ivreg2 z_large_donors (z_psu1919_vv z_interaction =z_foots_pop6m z_IV_interaction) z_elites `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A16.tex", dec(2) keep(z_psu1919_vv z_interaction) addstat("Number of clusters", e(N_clust), "Kleibergen-Paap F-stat", e(widstat)) nocons label noaster append
}
drop interaction IV_interaction z_interaction z_IV_interaction
***

*TAB_A17: ALTERNATIVE INSTRUMENTS
local urban "industry_workers_pop_m industry_firms_pop_m lit_m1911 elites bourgeoisie"
*PANEL A SPANISH FLU
foreach i of numlist 1 4 6 {
reg z_psu1919_vv z_excess_mf1918 `cntls_excess_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A17A.tex", dec(2) keep(z_excess_mf1918) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv{
foreach i of numlist 1 4 6 {
ivreg2 `depvar' (z_psu1919_vv=z_excess_mf1918) `cntls_excess_spec`i'', partial(province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A17A.tex", dec(2) keep(z_psu1919_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
}
*PANEL B RAINFALL
foreach i of numlist 1 4 6 {
ivreg2 z_psu1919_vv z_rain1919 z_rain_var `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921 cl1_stn1919)
outreg2 using "${outpath}/TAB_A17B.tex", dec(2) keep(z_rain1919) addstat("Number of district clusters", e(N_clust1),"Number of station clusters", e(N_clust2)) nocons label noaster append
}
*second stage
foreach depvar of varlist z_fascist_violence z_fascist_branch z_fascist1921_vv z_fascist1924_vv{
foreach i of numlist 1 4 6 {
ivreg2 `depvar' (z_psu1919_vv=z_rain1919) z_rain_var `cntls_spec`i'', partial(d_reg* province* z_lpop1911 z_lpop2 z_lpop3 z_lpop4) cl(circondario1921 cl1_stn1919)
outreg2 using "${outpath}/TAB_A17B.tex", dec(2) keep(z_psu1919_vv) addstat("Number of district clusters", e(N_clust1),"Number of station clusters", e(N_clust2), "1st stage F-stat", e(widstat)) nocons label noaster append
}
}

*TAB A18 FALSIFICATIONS RAINFALL 1917-1928
foreach year of numlist 1919 1918 1920/1928 {
*spec6
g instrument_year=z_rain`year'
la var instrument_year "Relative winter-spring rainfall"
ivreg2 z_fascist1924_vv instrument_year z_rain_var `cntls_spec6', partial(d_reg* province*) cl(circondario1921 cl1_stn`year')
outreg2 using "${outpath}/TAB_A18.tex", dec(2) keep(instrument_year) addstat("year",`year', "Number of district clusters", e(N_clust1),"Number of station clusters", e(N_clust2)) nocons label noaster append
drop instrument_year
}


*TAB_A19 ALTERNATIVE JEWS WITH INDEX - z_ave_fascism
***DUMMY
*PANEL A
foreach i of numlist 1/6 {
ivreg2 z_deportations_d (z_fascism_index=z_foots_pop6m) z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A19A.tex", dec(2) keep(z_fascism_index) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
***CAPPED AT 1
*PANEL B
foreach i of numlist 1/6 {
ivreg2 z_depo_pop_cap1 (z_fascism_index=z_foots_pop6m) z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'', partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A19B.tex", dec(2) keep(z_fascism_index) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
***CAPPED AT 1 NO CAMPS
*PANEL C
foreach i of numlist 1/6 {
ivreg2 z_depo_pop_cap1 (z_fascism_index=z_foots_pop6m) z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'' if KZ!=1, partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A19C.tex", dec(2) keep(z_fascism_index) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}

*TAB A20 JEWISH DEPORTATION -RF
***DUMMY
*PANEL A
foreach i of numlist 1/6 {
reg z_deportations_d z_foots_pop6m z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A20A.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
***CAPPED AT 1
*PANEL B
foreach i of numlist 1/6 {
reg z_depo_pop_cap1 z_foots_pop6m z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'', cl(circondario1921)
outreg2 using "${outpath}/TAB_A20B.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}
***CAPPED AT 1 NO CAMPS
*PANEL C1
foreach i of numlist 1/6 {
reg z_depo_pop_cap1 z_foots_pop6m z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'' if KZ!=1, cl(circondario1921)
outreg2 using "${outpath}/TAB_A20C.tex", dec(2) keep(z_foots_pop6m) addstat("Number of clusters", e(N_clust)) nocons label noaster append
}

*TAB_A21 JEWISH DEPORTATION SALO SAMPLE
***DUMMY
*PANEL A
foreach i of numlist 1/6 {
ivreg2 z_deportations_d (z_fascist1924_vv=z_foots_pop6m) z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'' if salo==1, partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A21A.tex", dec(2) keep(z_fascist1924_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
***CAPPED AT 1
*PANEL B
foreach i of numlist 1/6 {
ivreg2 z_depo_pop_cap1 (z_fascist1924_vv=z_foots_pop6m) z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'' if salo==1, partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A21B.tex", dec(2) keep(z_fascist1924_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
***CAPPED AT 1 NO CAMPS
*PANEL C
foreach i of numlist 1/6 {
ivreg2 z_depo_pop_cap1 (z_fascist1924_vv=z_foots_pop6m) z_jewish_pop1911 z_nazi_occ_duration `cntls_spec`i'' if salo==1 & KZ!=1, partial(d_reg* province*) cl(circondario1921)
outreg2 using "${outpath}/TAB_A21C.tex", dec(2) keep(z_fascist1924_vv) addstat("Number of clusters", e(N_clust), "1st stage F-stat", e(widstat)) nocons label noaster append
}
